CANADA

DEPARTMENT OF ENERGY, MINES AND RESOURCES
MINES BRANCH

COMPUTER PROGRAMS FOR X-RAY CRYSTALLOGRAPHY
PART II:
PROGRAM FOR DIFFRACTOMETER ANGLE SETTINGS

E. J. GABE

MINERAL SCIENCES DIVISION

JULY 1967

1867 1967

Research Report R 191 Price 50 cents

Available by mail from the Queen's Printer, Ottawa,
and at the following Canadian Government bookshops:

OTTAWA
Daly Building, Corner Mackenzie and Rideau

TORONTO
221 Yonge Street

MONTREAL
Æterna-Vie Building, 1182 St. Catherine St. West

WINNIPEG
Mall Center Building, 499 Portage Avenue

VANCOUVER
657 Granville Avenue

HALIFAX
1737 Barrington Street

or through your bookseller

A deposit copy of this publication is also available
for reference in public libraries across Canada

Price 50 cents Catalogue No. M38-1/191

Price subject to change without notice

ROGER DUHAMEL, F.R.S.C.
Queen's Printer and Controller of Stationery
Ottawa, Canada
1967

Mines Branch Research Report R 191

COMPUTER PROGRAMS FOR X-RAY CRYSTALLOGRAPHY. PART II: PROGRAM FOR DIFFRACTOMETER ANGLE SETTINGS

by

F. J. Gabe*

- -

ABSTRACT

This program calculates the three setting angles for a 4-circle diffractometer in the bisecting position ($\omega = o$). It is applicable to any system, and any type of systematic absence may be allowed for. The required angles may be calculated for any segment of reciprocal space.

---

*Research Scientist, Mineralogy Section, Mineral Sciences Division, Mines Branch, Department of Energy, Mines and Resources, Ottawa, Canada.

Direction des mines

Rapport de recherches R 191

PROGRAMMES D'ORDINATEUR POUR LA RADIOCRISTALLOGRAPHIE

PARTIE II: PROGRAMME POUR CALCULER LES ANGLES DE RÉGLAGE D'UN DIFFRACTOMÈTRE

par

E. J. Gabe*

- - -

RÉSUMÉ

Le présent programme calcule les trois angles de réglage pour un diffractomètre à 4 cercles en position bissectrice (ω = o). Il s'applique a tout système et l'on peut tenir compte de tout genre d'absence systématique. Les angles requis peuvent être calculés pour tout segment d'espace réciproque.

*Préposé à la recherche, Section de la minéralogie, Division des sciences minérales, Direction des mines, ministère de l'Énergie, des Mines et des Ressources, Ottawa, Canada.

CONTENTS

= = =

INTRODUCTION

This is the second program in the series of programs for X-ray crystallography. In order to use a 3- (or 4-) circle diffractometer for the collection of intensity data for structure analysis, it is necessary to calculate the setting angles ϕ, χ, (ω) and 2θ for every reflection h, k, ℓ within a sphere of radius $2 \sin\theta_{max}$. This program calculates the required angles for the $\omega = o$ case from orientation information about 3 reflections h_i, k_i, ℓ_i $(i = 1, 2, 3)$. The orientation matrix $\underset{\sim}{R}$ is calculated, as well as the real and reciprocal cell lengths and angles.

GENERAL DESCRIPTION

In essence, a 3- or 4-circle diffractometer is a device that will allow any reciprocal-space vector $\underset{\sim}{h}_i$ to be turned into a position so that diffraction can occur in a horizontal plane, i.e., the vector bisects the angle between the incident and diffracted beams. In a 3-circle device, this can only be done in two ways, for which $\omega = o$, but in a 4-circle device it may be done in an infinite number of ways. This allows the instrument to be set in a variety of ways, as well as permitting complete rotation about $\underset{\sim}{h}_i$.

The instrument is described in terms of the three Eulerian angles ω, χ, and ϕ. Reference to the illustration below will show how the Eulerian angles are chosen in this particular case:



If the θ-axial system is chosen so that Z_θ is parallel to the θ-axis of the instrument and Y_θ bisects the angle 2θ, then X_θ will be in the direction of the diffraction-vector. The other system is chosen so that Z_ϕ is along the ϕ-axis of the instrument and X_ϕ is at $\phi = 0$. Any vector x_ϕ, y_ϕ, z_ϕ may be brought into coincidence with the X_θ-axis by suitable changes to the angles ω, ϕ, and χ. To obtain the angles, consider a unit vector along X_θ:

The projection of x_θ onto OM $=$ $\cos\omega$

and x_θ onto ON $=$ $\sin\omega$.

The projection of OM onto OM$' = \cos\omega\cos\chi$,

and OM$'$ onto $X_\phi = \cos\omega\cos\chi\cos\phi$.

The projection of ON onto $X_\phi = -\sin\omega\sin\phi$;

thus, the projection of x_θ onto $X_\phi = \cos\omega\cos\chi\cos\phi - \sin\omega\sin\phi$.

Similarly,

the projection of OM onto $Y_\phi = \cos\omega\cos\chi\sin\phi$,

the projection of ON onto $Y_\phi = \sin\omega\cos\phi$;

thus the projection of x_θ onto $Y_\phi = \cos\omega\cos\chi\sin\phi + \sin\omega\cos\phi$.

Also,

the projection of OM onto $Z_\phi = \cos\omega\sin\chi$.

Hence, any vector in the direction X_θ is made up of components

$$\left.\begin{array}{l} \cos\omega\cos\chi\cos\phi - \sin\omega\sin\phi \\ \cos\omega\cos\chi\sin\phi + \sin\omega\cos\phi \\ \cos\omega\sin\chi \end{array}\right\} \quad (1)$$

in the ϕ-axis system.

Any vector in the crystal system (h, k, ℓ) may be described in terms of orthogonal components x_o, y_o, z_o if we define an orthogonality matrix T_{ij} such that

$$\begin{pmatrix} x_o \\ y_o \\ z_o \end{pmatrix} = \begin{pmatrix} T_{ij} \end{pmatrix} \begin{pmatrix} h \\ k \\ \ell \end{pmatrix},$$

where

$$T_{ij} = \begin{bmatrix} a^* & b^*\cos\gamma^* & c^*\cos\beta^* \\ o & b^*\sin\gamma^* & -c^*\sin\beta^*\cos\alpha \\ o & o & 1/c \end{bmatrix}$$

and x_o lies along a^*, y_o is in the a^*b^* plane and z_o is perpendicular to that plane.

There must also exist a rotation matrix R which will rotate the orthogonal system into coincidence with the ϕ-system, and we already know that if 3 angles ω, χ and ϕ are chosen so that the components of the vector in the ϕ-system satisfy (1), changes ω, χ, ϕ in the appropriate settings will bring the vector into the diffracting position, i.e. it will lie along x_θ. Thus,

$$\frac{2\sin\theta}{\lambda}\begin{pmatrix}\cos\omega\cos\chi\cos\phi - \sin\omega\sin\phi \\ \cos\omega\cos\phi\sin\phi + \sin\omega\cos\phi \\ \cos\omega\sin\chi\end{pmatrix} = \underset{\sim}{R}.\underset{\sim}{T}.\begin{pmatrix}h \\ k \\ \ell\end{pmatrix}, \qquad (2)$$

where $2\sin\theta/\lambda$ is the magnitude of the vector. If we have 3 non-coplanar reflections, h_i, k_i, ℓ_i, which define a right-handed coordinate system, we may write:

$$\underset{\sim}{\Theta} = \underset{\sim}{R}.\underset{\sim}{T}.\underset{\sim}{H},$$

where $\underset{\sim}{H}$ is the matrix of h, k, ℓ values and $\underset{\sim}{\Theta}$ is the matrix of angular components $\omega_i, \phi_i, \chi_i, \theta_i$. Thus,

$$\underset{\sim}{\Theta}\underset{\sim}{H}^{-1} = \underset{\sim}{R}.\underset{\sim}{T}.$$

Knowing $\underset{\sim}{T}$, it is possible to calculate $\underset{\sim}{R}$ but, in fact, this is not necessary. In the case of the 3-circle instrument, i.e. $\omega = 0$, for any reflection h, k, ℓ, (2) becomes:

$$\underset{\sim}{R}.\underset{\sim}{T}.\begin{pmatrix}h \\ k \\ \ell\end{pmatrix} = \begin{pmatrix}2\sin\theta/\lambda\cos\chi\cos\phi \\ 2\sin\theta/\lambda\cos\chi\sin\phi \\ 2\sin\theta/\lambda\sin\chi\end{pmatrix} = \begin{pmatrix}x_\phi \\ y_\phi \\ z_\phi\end{pmatrix}.$$

Hence, as we know the matrix $\underset{\sim}{R}.\underset{\sim}{T}.$,

$$\phi = \tan^{-1}\left(y_\phi/x_\phi\right)$$

$$\chi = \tan^{-1}\left(z_\phi\Big/\sqrt{x_\phi^2 + y_\phi^2}\right)$$

$$2\theta = 2\tan^{-1}\left(\sqrt{\left(x_\phi^2 + y_\phi^2 + z_\phi^2\right)\Big/\left(4 - x_\phi^2 - y_\phi^2 - z_\phi^2\right)}\right).$$

The 4-circle case is more easily treated with matrices. If we have a vector x_ϕ, y_ϕ, z_ϕ in the ϕ-system, it can be transformed to the Θ-system by applying a series of rotations Φ, X, Ω,

where

$$\underset{\sim}{\Phi} = \begin{bmatrix} \cos\phi & \sin\phi & o \\ -\sin\phi & \cos\phi & o \\ o & o & 1 \end{bmatrix},$$

$$\underset{\sim}{X} = \begin{bmatrix} \cos\chi & o & \sin\chi \\ o & 1 & o \\ -\sin\chi & o & \cos\chi \end{bmatrix},$$

$$\underset{\sim}{\Omega} = \begin{bmatrix} \cos\omega & \sin\omega & o \\ -\sin\omega & \cos\omega & o \\ o & o & 1 \end{bmatrix},$$

such that

$$\underset{\sim}{\Omega} \cdot \underset{\sim}{X} \cdot \underset{\sim}{\Phi} \cdot \begin{pmatrix} x_\phi \\ y_\phi \\ z_\phi \end{pmatrix} = \underset{\sim}{Q} \begin{pmatrix} x_\phi \\ y_\phi \\ z_\phi \end{pmatrix} = \begin{pmatrix} x_\phi \\ o \\ o \end{pmatrix}$$

if the Θ vector is in the diffracting position.

Hence,

$$\underset{\sim}{\Omega} \cdot \underset{\sim}{X} \cdot \underset{\sim}{\Phi} . \underset{\sim}{R} . \underset{\sim}{T} \begin{pmatrix} h \\ k \\ \ell \end{pmatrix} = \begin{pmatrix} x_\theta \\ o \\ o \end{pmatrix}$$

is the full equation for any reflection h, k, ℓ to diffract.

Expanding $\underset{\sim}{\Omega} . \underset{\sim}{X} . \underset{\sim}{\Phi}$ gives $\underset{\sim}{Q}$ =

$$\begin{bmatrix} \cos\omega\cos\chi\cos\phi - \sin\omega\sin\phi & \cos\omega\cos\chi\sin\phi + \sin\omega\cos\phi & \cos\omega\sin\chi \\ -\sin\omega\cos\chi\cos\phi - \cos\omega\sin\phi & -\sin\omega\cos\chi\sin\phi + \cos\omega\cos\phi & -\sin\omega\sin\chi \\ -\sin\chi\cos\phi & -\sin\chi\sin\phi & \cos\chi \end{bmatrix}.$$

From the expressions (1) we may find ϕ and χ for any value of ω we choose, and then form the matrix $\underset{\sim}{Q}$. It is more convenient, however, to consider only special cases ($\omega = 0$ or $\chi = 90$) where considerable simplification results. In any case, it is not particularly meaningful to find the settings χ and ϕ for arbitrary values of ω.

A more useful treatment is to consider rotation around the diffraction vector by the azimuthal angle ψ, for which the matrix $\underset{\sim}{Q}$ becomes:

$$\underset{\sim}{P}.\underset{\sim}{Q} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\psi & \sin\psi \\ 0 & -\sin\psi & \cos\psi \end{bmatrix} . \underset{\sim}{Q} \; ,$$

which still has the form of $\underset{\sim}{Q}$ and from which new values of ω, ϕ and χ may be extracted for any value of ψ:

$$\chi = \tan^{-1} \left(\sqrt{Q_{31}^2 + Q_{32}^2} \Big/ Q_{33} \right)$$
$$\phi = \tan^{-1} \left(Q_{32}/Q_{31} \right)$$
$$\omega = \tan^{-1} \left(-Q_{23}/Q_{13} \right)$$

If the starting position ($\psi = 0$) is arbitrarily chosen as $\omega = 0$, we have:

$$Q = P \begin{bmatrix} \cos\chi\cos\phi & \cos\chi\sin\phi & \sin\chi \\ -\sin\phi & \cos\phi & 0 \\ -\sin\chi\cos\phi & -\sin\chi\sin\phi & \cos\chi \end{bmatrix}$$

$$= \begin{bmatrix} \cos\chi\cos\phi & \cos\chi\sin\phi & \sin\chi \\ -\cos\psi\sin\phi - \sin\psi\sin\chi\cos\phi & \cos\psi\cos\phi - \sin\psi\sin\ \sin\phi & \sin\psi\cos\chi \\ \sin\psi\sin\phi - \cos\psi\sin\chi\cos\phi & -\sin\psi\cos\phi - \cos\psi\sin\chi\sin\phi & \cos\psi\cos\phi \end{bmatrix},$$

from which the values of ω, χ and ϕ may be extracted.

To return to the case of the 3 setting reflections, a useful by-product of the matrix $\underset{\sim}{R}.\underset{\sim}{T}$. is the metric tensor G^{-1}. This may be formed as follows:

$$G^{-1} = \underset{\sim}{R}.\underset{\sim}{T}.\underset{\sim}{R}.\underset{\sim}{T}$$
$$= \underset{\sim}{\tilde{T}}.\underset{\sim}{\tilde{R}}.\underset{\sim}{R}.T = \underset{\sim}{\tilde{T}}.\underset{\sim}{T}$$

because $\underset{\sim}{R}$ is orthogonal. Then

$$G^{-1}_{ij} = a^*_i a^*_j \cos\alpha^*_{ij} ,$$

from which the reciprocal cell parameters may be found. In similar manner the direct cell parameters may be found from G.

The separate problem of indexing through any segment of reciprocal space with any order of changes to h, k, and ℓ, has been solved as follows: Any reflection $hk\ell$ (referred to hereafter as h) may be reached from an origin reflection h_0 by an integral number of steps Δh in the 3 directions; i.e.

$$h = h_0 + n_1\Delta h_1 + n_2\Delta h_2 + n_3\Delta h_3 ,$$

i.e.
$$\begin{pmatrix} h \\ k \\ \ell \end{pmatrix} = \begin{pmatrix} h_o \\ k_o \\ \ell_o \end{pmatrix} + \begin{bmatrix} \Delta h_1 & \Delta h_2 & \Delta h_3 \\ \Delta k_1 & \Delta k_2 & \Delta k_3 \\ \Delta \ell 1 & \Delta \ell 2 & \Delta \ell 3 \end{bmatrix} \begin{pmatrix} n_1 \\ n_2 \\ n_3 \end{pmatrix},$$

$$\text{or } (h_i) = (h_{oi}) + \left[\Delta h_{ij}\right] (n_j) .$$

The steps Δh_{i1}, Δh_{i2} and Δh_{i3} are chosen so that they are increments in going from one layer to the next, one line to the next, and one point to the next, in reciprocal space.

If we wish to have layers normal to h, lines parallel to k, and successive points along ℓ,

$$\left[\Delta h_{ij}\right] = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}.$$

Thus, the 3 steps Δh represent the 3 directions formed by the intersections of the 3 planes bounding the required volume of reciprocal space. It is easy to see how this can be adapted to cover the limited regions required in the higher-symmetry systems. It may be necessary, or more convenient, to scan the required volume of reciprocal space in more than one segment but at the same time avoid duplicating reflections already considered or symmetrically equivalent reflections. This is achieved by allowing a different choice of origin reflection for each segment.

In order to proceed from line to line and layer to layer correctly, it is necessary to know the starting reflection for each line and layer. This is worked out at the beginning of each segment, by specifying a starting reflection h_s and then working out the first line and first layer reflections as follows:

$$(h_{si}) = (h_{oi}) + \left[\Delta h_{ij}\right] (n_j);$$

therefore $$\left[\Delta h_{ij}\right]^{-1} (h_{si} - h_{oi}) = (n_j). \qquad (3)$$

From which the first reflection of the layer (subscript L) is:

$$h_{Li} = h_{oi} + \left[\Delta h_{i1}\right] n_1 ,$$

the first reflection of the line (subscript ℓ) is:

$$h_{\ell i} = h_{Li} + \left[\Delta h_{i2}\right] n_2 , \qquad (4)$$

and the first point (subscript s) is:

$$h_{si} = h_{\ell i} + \left[\Delta h_{i3}\right] n_3 .$$

An example will clarify the procedure. Suppose we wish to find all the reflections with h negative for a monoclinic crystal, starting at the reflection -3, 2, 4 and indexing so that k changes fastest, h next and ℓ least.

$$\Delta h_{ij} = \begin{bmatrix} 0 & -1 & 0 \\ 0 & 0 & 1 \\ 1 & 0 & 0 \end{bmatrix} \text{ and } \left[\Delta h_{ij}\right]^{-1} = \begin{bmatrix} 0 & 0 & 1 \\ -1 & 0 & 0 \\ 0 & 1 & 0 \end{bmatrix}.$$

h^{0i} is -1, 0, 1 as we are considering a monoclinic crystal and we do not wish to repeat reflections with h = 0 or ℓ = o. Thus, from (3):

$$\left[\Delta h_{ij}\right]^{-1} \begin{pmatrix} -3 & - & -1 \\ 2 & - & 0 \\ 4 & - & 1 \end{pmatrix} = \begin{bmatrix} 0 & 0 & 1 \\ -1 & 0 & 0 \\ 0 & 1 & 0 \end{bmatrix} \begin{pmatrix} -2 \\ 2 \\ 3 \end{pmatrix} = \begin{pmatrix} 3 \\ 2 \\ 2 \end{pmatrix} = (n_j),$$

and from (4):

$$\left.\begin{aligned} h_L &= -1 + 0 = -1 \\ k_L &= 0 + 0 = 0 \\ \ell_L &= 1 + 3 = 4 \end{aligned}\right\} \text{ Start of layer,}$$

$$\left.\begin{aligned} h_\ell &= -2 + -1 = -3 \\ k_\ell &= 0 + 0 = 0 \\ \ell_\ell &= 0 + 4 = 4 \end{aligned}\right\} \text{ Start of line within layer,}$$

$$\left.\begin{aligned} h_s &= 0 + -3 = -3 \\ k_s &= 2 + 0 = 2 \\ \ell_s &= 0 + 4 = 4 \end{aligned}\right\} \text{ Starting or current reflection.}$$

After each reflection the current values are incremented until a limit is exceeded. The line reflection is then incremented and the current reflection set equal to it, until a limit is exceeded. The layer reflection is then incremented and the line and current reflections set equal to it until a limit is exceeded.

DETAILS OF THE CALCULATION

h, k, ℓ, ϕ, χ, θ values are read for 3 reflections which define a right-handed coordinate system. From the details of these reflections the matrix $\underset{\sim}{R}.\underset{\sim}{T}$ is computed and, from this, G^{-1} and G, which give the reciprocal and real cell parameters. Further information concerning the segments of reciprocal space required and the order of indexing is read together with systematic absence information. Each reflection h, k, ℓ is tested against an expression of the type:

$$Ah + Bk + C\ell = Dm + E,$$

where A, B, C, D and E determine if the reflection is acceptable. Any number of these conditions may be applied to axial, zonal or general reflections.

From the details of the starting and origin reflections given, n_1, n_2 and n_3 are calculated, and then reflections are generated in the preset sequence until all reflections in the reciprocal segment with 2θ values less than some limit have been dealt with. The same operations are then repeated for any other segments required.

Input:

The details of the description of both input and output are the same as for the first program in this series. The input is all from punched cards and the output is all on the lineprinter.

Card Column (c.c.)	Contents	Comments
First Card		
1-3	xxx	No. of structures to be processed
4-80	blanks	

The remainder of the cards are in sets, 1 set per structure.

Card Column (c.c.)	Contents	Comments
Title Card		
1-70	Title of structure	
71-80	blanks	
Wavelength Card		
1-8	xx.xxxxx	Wavelength of radiation used
9-80	blanks	

Reflection Cards

There are three of these cards.

Card Column (c.c.)	Contents	Comments
1-4	xxx b	h
5-8	xxx b	k
9-12	xxx b	ℓ
13-21	xxx.xxxx b	ϕ } Values for reflection h, k, ℓ
22-30	xxx.xxxx b	χ } Values for reflection h, k, ℓ
31-39	xxx.xxxx b	θ } Values for reflection h, k, ℓ
40-80	blanks	

Limits Card

1-4	xxx b	No. of reciprocal space segments (N)
5-10	xxx.xx	Maximum value of 2θ
11-80	blanks	

Reflection Condition Cards

1-3	xxx	No. of reflection conditions (R)
4-80	blanks	

This is followed by R cards

1-5	xxx bb	Type of reflection to which condition applies; see below
6-10	xxx bb	A
11-15	xxx bb	B
16-20	xxx bb	C
21-25	xxx bb	D
26-30	xxx bb	E
31-80	blanks	

The first number has the value 1-7 as follows:

1	00ℓ	reflections only
2	$0k0$	" "
3	$h00$	" "
4	$0k\ell$	" "
5	$h0\ell$	" "
6	$hk0$	" "
7	$hk\ell$	" "

e.g., parameters 7 1 1 1 2 0 would mean that for all reflections h, k, ℓ,

$$Ah + Bk + C\ell = Dn + E$$

would be

$$h + k + \ell = 2n$$

for the reflection to be present.

Any number R of these cards may be given, and multiple conditions may be given for the same reflection type.

Segment Cards

The reflection condition cards are followed by N-cards, each of which deals with the indexing of 1 segment of reciprocal space.

1-4	xxx b	h_o	
5-8	xxx b	k_o	Origin-defining reflection
9-12	xxx b	ℓ_o	
13-16	xxx b	h_{11}	
17-20	xxx b	h_{21}	Increment steps in $hk\ell$ for layers
21-24	xxx b	h_{31}	
25-28	xxx b	h_{12}	
29-32	xxx b	h_{22}	Increment steps in $hk\ell$ for lines
33-36	xxx b	h_{32}	
37-40	xxx b	h_{13}	
41-44	xxx b	h_{23}	Increment steps in $hk\ell$ for points
45-48	xxx b	h_{33}	
49-52	xxx b	h_s	
53-56	xxx b	k_s	Starting reflection
57-60	xxx b	ℓ_s	
61-80	blanks		

Output:

The first page of output gives the matrix $\underset{\sim}{R}.\underset{\sim}{T}$ from which all calculations are done, the reciprocal and real lattice parameters, and the maximum values of h, k, ℓ. The subsequent pages list $h, k, \ell, \phi, \chi, 2\theta$ and Lp^{-1} for all reflections. All output quantities are labelled and no explanation is needed.

This whole sequence is repeated as many times as there are structures to be processed.

REFERENCES

William R. Busing and Henri A. Levy. Angle Calculations for 3- and 4-circle X-ray and Neutron Diffractometers. Acta Cryst. 22, 457 (1967).

J. S. Rollett. Computing Methods in Crystallography. Pergamon Press, Oxford, London, Edinburgh, New York and Frankfurt (1965).

Frank Ayres Jr. Theory and Problems of Matrices. Schaum Publishing Co., New York (1962).

- - -

EJG:(PFS) vb

PROGRAM LISTING

FORTRAN IV G LEVEL 0, MOD 0

MAIN

```
C PROGRAM TO CALCULATE SETTING ANGLES PHI,CHI AND TWOTHETA FOR ALL PERMISSIBLE
C REFLECTIONS FOR ANY SYSTEM. GIVEN H,K,L,PHI,CHI AND THETA FOR ANY THREE
C NON-COPLANAR REFLECTIONS WHICH DEFINE A RIGHT HANDED SYSTEM
      DIMENSION H(3),K(3),L(3),PHA(3),CHA(3),THATE(3),V(3,3),STHETA(3),
     1CPHI(3),SPHI(3),CCHI(3),SCHI(3),TRG(3,3),VI(3,3),R(3,3),GI(3,3),
     2A(3),ANG(3),G(3,3),AS(3),ANGS(3),FDH(3,3),FDHI(3,3),FSTHKL(3,3),
     3VEC(3),VECVEC(3),DH(3,3),IND(3),RT(3,3),ICOND(10),HS(10),KS(10),
     4LS(10),IR(10),IS(10),INDEX(3)
      INTEGER H,DH,X,Y,Z,HO,HS,HMAX,FSTHKL
      LNCD=1
      LNPT=3
      READ(LNCD,9)NSTRUC
      DO 1000 NSETT=1,NSTRUC
      DEGREE=180.0/3.141593
      RADIAN=1.0/DEGREE
      WRITE(LNPT,3)
 3    FORMAT(1H1,/)
      READ(LNCD,13)
      WRITE(LNPT,13)
 13   FORMAT(70H
     1                    )
      READ(LNCD,1)WAVE
 1    FORMAT(F8.5)
      DO 100 I=1,3
      READ(LNCD,2)H(I),K(I),L(I),PHA(I),CHA(I),THATE(I)
 2    FORMAT(3(I3,1X),3(F8.4,1X))
      CPHI(I)=COS(PHA(I)*RADIAN)
      SPHI(I)=SIN(PHA(I)*RADIAN)
      CCHI(I)=COS(CHA(I)*RADIAN)
      SCHI(I)=SIN(CHA(I)*RADIAN)
      STHETA(I)=2.0/WAVE*SIN(THATE(I)*RADIAN)
      V(3,I)=L(I)
      V(2,I)=K(I)
 100  V(1,I)=H(I)
C TRIG MATRIX HPHI
      DO 101 I=1,3
      TRG(1,I)=STHETA(I)*CCHI(I)*CPHI(I)
      TRG(2,I)=STHETA(I)*CCHI(I)*SPHI(I)
 101  TRG(3,I)=STHETA(I)*SCHI(I)
C INVERT INDEX MATRIX
      CALL INVERT (V,VI)
C COMPUTE UB MATRIX %R< AND PRINT
      CALL MATMUL (TRG,VI,R)
 4    FORMAT(/10X,2HR%,I1,5H,1<# ,F12.8,4X,2HR%,I1,5H,2<# ,F12.8,4X,
     12HR%,I1,5H,3<# ,F12.8)
      DO 103 I=1,3
 103  WRITE(LNPT,4)I,R(I,1),I,R(I,2),I,R(I,3)
C  COMPUTE RECIPROCAL AND REAL METRIC TENSORS G-1 AND G
```

```
      DO 104 I=1,3
      DO 104 J=1,3
 104  RT(J,I)=R(I,J)
      CALL MATMUL (RT,R,GI)
      CALL PARAM (GI,AS,ANGS)
      WRITE(LNPT,5)AS(1),AS(2),AS(3),ANGS(1),ANGS(2),ANGS(3)
  5   FORMAT(///5X,4HA*# ,F10.6,5X,4HB*# ,F10.6,5X,4HC*# ,F10.6,5X,
     16HALF*# ,F10.5,5X,6HBET*# ,F10.5,5X,6HGAM*# ,F10.5)
      CALL INVERT (GI,G)
      CALL PARAM (G,A,ANG)
      WRITE(LNPT,6)A(1),A(2),A(3),ANG(1),ANG(2),ANG(3)
  6   FORMAT(///5X,4HA # ,F10.6,5X,4HB # ,F10.6,5X,4HC # ,F10.6,5X,
     16HALF # ,F10.5,5X,6HBET # ,F10.5,5X,6HGAM # ,F10.5)
      DO 105 I=1,3
      DO 105 J=1,3
 105  R(I,J)=R(I,J)*WAVE
C                 ***** SETTINGS PROCEDURE *****
C READ NUMBER OF RECIPROCAL SEGMENTS ANGULAR LIMITS AND ABSENCE CODES
      READ(LNCD,8)NUMSEG,THETA2
      S=SIN(RADIAN*THETA2*0.5)
      SS2=S+S
      SS4=SS2*SS2
      DO 200 J=1,3
      ANG(J)=SIN(ANG(J)*RADIAN)
 200  ANGS(J)=SIN(ANGS(J)*RADIAN)
      HMAX=SS2/(AS(1)*ANGS(2)*ANG(3)*WAVE)+1.0
      KMAX=SS2/(AS(2)*ANGS(3)*ANG(1)*WAVE)+1.0
      LMAX=SS2/(AS(3)*ANGS(1)*ANG(2)*WAVE)+1.0
      WRITE(LNPT,14)HMAX,KMAX,LMAX
 14   FORMAT(///,3(6X,I3))
C READ REFLECTION CONDITIONS
      READ(LNCD,9)NCOND
  9   FORMAT(I3)
  8   FORMAT(I3,1X,F6.2)
      IF(NCOND.EQ.0) GO TO 208
      DO 207 I=1,NCOND
 207  READ(LNCD,10)ICOND(I),HS(I),KS(I),LS(I),IR(I),IS(I)
 10   FORMAT(6(I3,2X))
 208  DO 1000 NSEG=1,NUMSEG
      READ(LNCD,7)HO,KO,LO,DH(1,1),DH(2,1),DH(3,1),DH(1,2),DH(2,2),
     1DH(3,2),DH(1,3),DH(2,3),DH(3,3),IND(1),IND(2),IND(3)
  7   FORMAT(15(I3,1X))
      DO 201 I=1,3
      DO 201 J=1,3
 201  FDH(I,J)=DH(I,J)
      CALL INVERT (FDH,FDHI)
      DO 202 I=1,3
      INDEX(I)=FDHI(I,1)*(IND(1)-HO)+FDHI(I,2)*(IND(2)-KO)+FDHI(I,3)
```

```
     1*(IND(3)-L0)
      IF(INDEX(I).GE.0) GO TO 205
      INDEX(I)=INDEX(I)-0.5
      GO TO 202
 205  INDEX(I)=INDEX(I)+0.5
 202  CONTINUE
      FSTHKL(1,1)=DH(1,1)*INDEX(1)+H0
      FSTHKL(2,1)=DH(2,1)*INDEX(1)+K0
      FSTHKL(3,1)=DH(3,1)*INDEX(1)+L0
      DO 203 I=1,3
      FSTHKL(I,2)=DH(I,2)*INDEX(2)+FSTHKL(I,1)
 203  FSTHKL(I,3)=DH(I,3)*INDEX(3)+FSTHKL(I,2)
C START OF REFLECTION
      NPAGE=55
 301  X=FSTHKL(1,3)
      Y=FSTHKL(2,3)
      Z=FSTHKL(3,3)
C  START A NEW PAGE AND PRINT HEADINGS
      GO TO 503
 302  NPAGE=0
      WRITE(LNPT,11)
 11   FORMAT(1H1,//20X,1HH,4X,1HK,4X,1HL,5X,7H  PHI  ,5X,7H  CHI  ,5X,7H
     1 2THETA,5X,6H LP-1 ,/)
 303  IF(X.EQ.0.AND.Y.EQ.0.AND.Z.EQ.0) GO TO 500
      IF(NCOND.EQ.0) GO TO 304
      DO 300 N=1,NCOND
      JCOND=ICOND(N)
      GO TO(310,320,330,340,350,360,370),JCOND
 310  IF(X.EQ.0.AND.Y.EQ.0) GO TO 370
      GO TO 300
 320  IF(X.EQ.0.AND.Z.EQ.0) GO TO 370
      GO TO 300
 330  IF(Y.EQ.0.AND.Z.EQ.0) GO TO 370
      GO TO 300
 340  IF(X.EQ.0) GO TO 370
      GO TO 300
 350  IF(Y.EQ.0) GO TO 370
      GO TO 300
 360  IF(Z.EQ.0) GO TO 370
      GO TO 300
 370  LHS=IABS(X*HS(N)+Y*KS(N)+Z*LS(N))
      M=IR(N)
      IF(MOD(LHS,M).NE.IS(N)) GO TO 500
 300  CONTINUE
C PHI,CHI AND 2THETA CALCULATION
 304  SIGMA=0.0
      DO 400 I=1,3
      VEC(I)=R(I,1)*X+R(I,2)*Y+R(I,3)*Z
```

```
 400   SIGMA=SIGMA+VEC(I)*VEC(I)
       IF(SIGMA.GE.SS4) GO TO 500
       BOT=ABS(VEC(1))
       CEN=ABS(VEC(2))
       TOP=ABS(VEC(3))
       IF(BOT.NE.0.0) PHI=ATAN2(CEN,BOT)*DEGREE
       IF(BOT.EQ.0.0) PHI=90.0
       SIGMA=SIGMA-TOP*TOP
       IF(SIGMA.NE.0.0) CHI=ATAN2(TOP,SQRT(SIGMA))*DEGREE
       IF(SIGMA.EQ.0.0) CHI=90.0
CPHI AND CHI IN 270 TO 90 RANGE
       IF(VEC(3).LT.0.0) CHI=360.0-CHI
       IF(VEC(1).LT.0.0) GO TO 401
       IF(VEC(2).LT.0.0) PHI=360.0-PHI
       GO TO 402
 401   IF(VEC(2).LT.0.0) PHI=180.0+PHI
       IF(VEC(2).GE.0.0) PHI=180.0-PHI
 402   IF(CHI.EQ.90.0.OR.CHI.EQ.270.0) PHI=999.0
       SINSQ=0.25*(SIGMA+TOP*TOP)
       THETA=2.0*DEGREE*ATAN(SQRT(SINSQ/(1.0-SINSQ)))
       TOP=4.0*SQRT(SINSQ*(1.0-SINSQ))
       BOT=1.0-SINSQ-SINSQ
       BOT=1.0+BOT*BOT
       POL=TOP/BOT
       WRITE(LNPT,12)X,Y,Z,PHI,CHI,THETA,POL
       NPAGE=NPAGE+1
 12    FORMAT(18X,I3,2X,I3,2X,I3,3(5X,F7.3),5X,F6.4)
C INCREMENT INDICES
 500   X=X+DH(1,3)
       Y=Y+DH(2,3)
       Z=Z+DH(3,3)
       IX=IABS(X)
       IY=IABS(Y)
       IZ=IABS(Z)
       IF(IX.LT.HMAX.AND.IY.LT.KMAX.AND.IZ.LT.LMAX) GO TO 503
       DO 501 I=1,3
       FSTHKL(I,2)=FSTHKL(I,2)+DH(I,2)
 501   FSTHKL(I,3)=FSTHKL(I,2)
       IX=IABS(FSTHKL(1,3))
       IY=IABS(FSTHKL(2,3))
       IZ=IABS(FSTHKL(3,3))
       IF(IX.LT.HMAX.AND.IY.LT.KMAX.AND.IZ.LT.LMAX) GO TO 301
       DO 502 I=1,3
       FSTHKL(I,1)=FSTHKL(I,1)+DH(I,1)
       FSTHKL(I,2)=FSTHKL(I,1)
 502   FSTHKL(I,3)=FSTHKL(I,2)
       IX=IABS(FSTHKL(1,3))
       IY=IABS(FSTHKL(2,3))
```

```
      IZ=IABS(FSTHKL(3,3))
      IF(IX.LT.HMAX.AND.IY.LT.KMAX.AND.IZ.LT.LMAX) GO TO 301
      GO TO 1000
503   IF(NPAGE.GT.54) GO TO 302
      GO TO 303
1000  CONTINUE
      STOP
      END
```

SUBROUTINES

```
C INVERT 3X3 MATRIX U TO GIVE UI
      SUBROUTINE INVERT (U,UI)
      DIMENSION U(3,3),UI(3,3)
      UI(1,1)=U(2,2)*U(3,3)-U(2,3)*U(3,2)
      UI(2,1)=-(U(2,1)*U(3,3)-U(2,3)*U(3,1))
      UI(3,1)=U(2,1)*U(3,2)-U(2,2)*U(3,1)
      UI(1,2)=-(U(1,2)*U(3,3)-U(1,3)*U(3,2))
      UI(2,2)=U(1,1)*U(3,3)-U(1,3)*U(3,1)
      UI(3,2)=-(U(1,1)*U(3,2)-U(1,2)*U(3,1))
      UI(1,3)=U(1,2)*U(2,3)-U(1,3)*U(2,2)
      UI(2,3)=-(U(1,1)*U(2,3)-U(1,3)*U(2,1))
      UI(3,3)=U(1,1)*U(2,2)-U(1,2)*U(2,1)
      DMAT=U(1,1)*UI(1,1)+U(1,2)*UI(2,1)+U(1,3)*UI(3,1)
      DO 1000 I=1,3
      DO 1000 J=1,3
 1000 UI(I,J)=UI(I,J)/DMAT
      RETURN
      END


C MULTIPLY TWO MATRICES TOGETHER
      SUBROUTINE MATMUL (AMAT,BMAT,CMAT)
      DIMENSION AMAT(3,3),BMAT(3,3),CMAT(3,3)
      DO 2000 I=1,3
      DO 2000 J=1,3
      CMAT(I,J)=0.0
      DO 2000 K=1,3
 2000 CMAT(I,J)=CMAT(I,J)+AMAT(I,K)*BMAT(K,J)
      RETURN
      END


C EXTRACT ELEMENTS FROM METRIC TENSOR
      SUBROUTINE PARAM (W,D,T)
      DIMENSION W(3,3),D(3),T(3)
      DO 3000 I=1,3
      J=I
 3000 D(I)=SQRT(W(I,J))
      T(1)=57.29578*ARCOS(W(2,3)/(D(2)*D(3)))
      T(2)=57.29578*ARCOS(W(1,3)/(D(1)*D(3)))
      T(3)=57.29578*ARCOS(W(1,2)/(D(1)*D(2)))
      RETURN
      END
```

==